
Come ci parla l'SQL?

SQLCA
Communication Area

SQLDA
Descriptor Area

SQLDA – SQL Descriptor Area

 E' un insieme di variabili utilizzate nell'istruzione describe
oppure dalle istruzioni prepare, open, call, fetch e 
execute

 In RPG l'SQLDA è definita dal precompilatore SQL

 L'SQLDA utilizzata in describe o prepare restituisce 
informazioni sull'istruzione select che è stata preparata

Manipolare il descrittore

 A partire dalla V5R4 il DB2 agevola l'utilizzo della struttura SQLDA
tramite l'uso dell'SQL descriptor

 allocate descriptor, deallocate descriptor

 get descriptor, set descriptor reperisce o imposta le 
informazioni di un descrittore

 using sql descriptor clausola da aggiungere su istruzione open 
per utilizzare il descrittore precedentemente impostato

 into sql descriptor clausola da aggiungere su istruzione fetch
per indirizzare l'ouput al descrittore precedentemente impostato

 describe istruzione per reperire le informazioni di definizione di una 
tabella o del resultset di un'istruzione sql di select

 queste istruzioni possono essere utilizzate solo in embedded SQL

Allocazione, impostazione e lettura descrittore

Resultset variabile/1

 Un caso tipico in cui il descrittore SQL ci viene in aiuto con 
un'istruzione costruita dinamicamente è quando l'elenco delle 
colonne del resultset è variabile

 quante variabili host ho bisogno per la clausola INTO?

 come definisco le variabili host in modo che siano congruenti con il tipo dati 
del resultset?

Ufficio rif.:    ___
Sequenza colonne: Codice 0 Descrizione 1 Uff.rif. 3 Resp. 2
(0 per non visualizzare)

xxxxxxxxxxxxxxx xxx xxxxxx
xxxxxxxxxxxxxxx xxx xxxxxx
. . .

select DEPTNAME, MGRNO, ADMRDEPT

from DEPARTMENT

where ADMRDEPT = ?

order by DEPTNO;

filtro sui record

ordine in cui 
visualizzare le 

colonne del 
resultset



Resultset variabile/2

 allocazione del descrittore

 costruisco la clausola select in base alle scelte dell'utente. Il codice 
può essere più o meno complesso in base a come viene proposto 
all'utente la possibilità di scelta (elenco di campi preimpostato di 
cui scegliere l'ordine, digitazione libera dei nomi campi, ecc.)

 a cui aggiungo le clausole from e where

exec sql

allocate descriptor local :Desc with max 4;

sqlstm = 'select DEPTNAME, MGRNO, DEPTNO';

sqlstm = sqlstm + ' from DEPARTMENT +

where ADMRDEPT = ?';

dcl-s sqlstm

varchar(500);

dcl-s Desc varchar(20) 

inz('dCurs1');

Esempio allegato: 
DESC02.SQLRPGLE



Resultset variabile/3

 Una volta costruita nella variabile sqlstm l'istruzione SQL eseguo 
la preparazione

 Tramite il descrittore reperisco le informazioni sull'istruzione 
appena preparata

 Reperisco il numero di colonne del resultset

exec sql

prepare pCurs1 from :sqlstm;

exec sql

describe pCurs1 using sql descriptor local :Desc;

exec sql

get descriptor :Desc :numCol = COUNT;

dcl-s numCol int(10);

exec sql declare Curs1

cursor for pCurs1;SQLDA

intestazione

descrizione 
colonna 1

descrizione 
colonna 2

descrizione 
colonna …



Resultset variabile/4

 Apro il cursore

 Leggo il cursore indirizzando l'output al descrittore SQL

 Per ogni colonna leggo dati e informazioni dal descrittore

exec sql

open Curs1 using :WSDEPT;

exec sql

fetch next from Curs1 into sql descriptor :Desc;

// reperimento nome della prima colonna della tabella

exec sql

get descriptor :Desc value 1 :MyCol = DB2_SYSTEM_COLUMN_NAME;

// reperimento dati della prima colonna della tabella

exec sql

get descriptor :Desc value 1 :MyVar = DATA;
ATTENZIONE all'esatta 

corrispondenza del tipo datisqlcode = -76



Resultset variable/5

 Altre informazioni reperibili dal descrittore con get
descriptor:

 header:

DB2_CURSOR_UPDATABILITY: 1 se cursore aggiornabile

 item:

TYPE, LENGTH, PRECISION, SCALE: tipo dati, lunghezza, precisione e scala 
della colonna

DATETIME_INTERVAL_CODE: specificazione del tipo dati data (1 date, 2
time, 3 timestamp)

DB2_COLUMN_UPDATABILITY: 1 se la colonna è aggiornabile

DB2_LABEL: etichetta della colonna

type
1 = character
12 = varchar
2 = numeric
3 = decimal
4 = integer
5 = smallint
9 = date/time



Parameter markers variabili + resultset variabile?/1

 Nell'esempio del Power coffee 9/2022 abbiamo utilizzato il 
descrittore per governare il numero variabile di parameter markers

 Ora abbiamo utilizzato il descrittore per poter gestire un resulset
variabile

 E' possibile combinare le due caratteristiche per gestire un SQL 
dinamico che abbia un numero variabile di parameter markers e un 
numero variabile di colonne nel resultset?

Parameter markers variabili + resultset variabile?/2

 riprendiamo l'esempio visto in precedenza aggiungendo la 
possibilità che il valore blanks nel campo "Ufficio rif." indichi di non 
filtrare i records

Ufficio rif.:    ___ (blanks = tutti)
Sequenza colonne: Codice 0 Descrizione 1 Uff.rif. 3 Resp. 2
(0 per non visualizzare)

xxxxxxxxxxxxxxx xxx xxxxxx
xxxxxxxxxxxxxxx xxx xxxxxx
. . .

select DEPTNAME, MGRNO, ADMRDEPT

from DEPARTMENT

where ADMRDEPT = ?

order by DEPTNO;

filtro sui record
opzionale

ordine in cui 
visualizzare le 

colonne del 
resultset



Parameter markers variabili + resultset variabile?/3

 allocazione di due descrittori

 se uso il parameter marker imposto il descrittore dCurs1A

 altrimenti imposto il numero elementi a 0

Esempio allegato: 
DESC03.SQLRPGLE

// descrittore per i parameter markers

exec sql

allocate descriptor local :DescA with max 1;

// descrittore per il resultset

exec sql

allocate descriptor local :DescB with max 4;

dcl-s DescA varchar(20) 

inz('dCurs1A');

dcl-s DescB varchar(20) 

inz('dCurs1B');

exec sql

set descriptor :descA

value 1 type = 1, length = 3, data = :pmDept;

exec sql

set descriptor :descA count = 1;

exec sql

set descriptor :descA count = 0;



Parameter markers variabili + resultset variabile?/4

 l'altro descrittore dCurs1B lo utilizzo per reperire le informazioni 
sul resultset

 per l'apertura del cursore uso il descrittore dCurs1A

 per la lettura del cursore uso il descrittore dCurs1B

exec sql

prepare pCurs1 from :sqlstm;

exec sql

describe pCurs1 using sql descriptor local :DescB;

exec sql

get descriptor :DescB :numCol = COUNT;

exec sql

open Curs1 using SQL descriptor :descA;

exec sql

fetch next from Curs1 into sql descriptor :DescB;



Riferimenti

 E-mail aziendale: mriva@sirio-is.it

 Blog: www.markonetools.it

 E-mail blog: info@markonetools.it

 Linkedin: www.linkedin.com/in/marcoriva-mk1

 Twitter: @MarcoRiva73

 Facebook: https://www.facebook.com/markonetools/

 YouTube: https://www.youtube.com/channel/UCb47YJQJCzU-5x4nnGzDu-w

**free

// ESEMPIO UTILIZZO SQL DESCRIPTOR CON SQL DINAMICO CON RESULTSET VARIABILE
// (c) MarkOne 2022, www.markonetools.it

// esempio di costruzione di una istruzione SQL dinamica in base a un
//  parametro di filtro fisso (codice ufficio di riferimento)
//  ma con output del resultset variabile per numero e ordine delle colonne
// per semplificare l'esempio sono state omesse tutte le parti di controllo
//  dell'esito delle istruzioni SQL
// in questo esempio viene costruito in output un file csv

// esempio utilizzo:
//   call desc02 parm('A00' '0132')
//    parm n.1: codice ufficio di riferimento per filtrare i record della
//              tabella DEPARTMENT
//    parm n.2: ordine delle 4 colonne del resultset (DEPTNO, DEPTNAME, MGRNO, ADMRDEPT)
//              '0' per non visualizzare la colonna

// Curs1: cursore sql
// dCurs1: descrittore sql (referenziato con variabile Desc)
// pCurs1: istruzione preparata per il cursore Curs1 in base alla stringa sqlstm

//--- MAIN INTERFACE ---
dcl-pr DESC02 extpgm('DESC02');
 *n like(DEPTNO);
 *n char(4);
end-pr;
dcl-pi DESC02;
 pmDept like(DEPTNO);
 pmOrd char(4);
end-pi;

dcl-ds DEPARTMENT ext template end-ds;

// istruzione SQL dinamica
dcl-s sqlstm varchar(1000);
// numero massimo di elementi usati nel descrittore
dcl-s MaxEl uns(3) inz(%size(pmOrd));

// array per ordinamento colonne del resultset (parametro input)
dcl-ds OrdCol dim(%size(pmOrd)) qualified;
  Ord zoned(1);
  Col char(10);
end-ds;
// intestazioni colonne csv output
dcl-s NomeCol varchar(10);
// tipo dati della colonna di output
dcl-s TypeCol int(10);
// contenuto della colonna di output
dcl-s wVarChar varchar(100);
dcl-s wChar char(100);
// numero di colonne del resultset
dcl-s numCol uns(3);
dcl-s i uns(3);

// nome descrittore
dcl-s Desc varchar(20) inz('dCurs1');

// boolean per primo record letto dal resultset
dcl-s firstRow ind inz(*on);

// percorso file csv di output
dcl-s FileCsv varchar(256) inz('/home/MK1sample/Department.csv');
// riga di intestazione del csv
dcl-s csvInt varchar(500);
// altre righe del csv
dcl-s csvRiga varchar(500);

// dichiarazione cursore
exec sql
  declare Curs1 cursor for pCurs1;

//------------------------------------------------------------------------------

// allocazione descrittore SQL con massimo 4 elementi
//  ovverosia il numero massimo di colonne del resultset
exec sql
  allocate descriptor local :Desc with max :MaxEl;

// -> array popolato in base a input dell'utente
// OrdCol indice 1        2          3       4
// .Col          DEPTNO _ DEPTNAME _ MGRNO _ ADMRDEPT _
// .Ord                 0          1       3          2  <-- ordinamento scelto da utente
// -> Riordino l'array in base a .Ord
// OrdCol indice 1        2          3       4
// .Col          DEPTNAME _ ADMRDEPT _ MGRNO _
// .Ord                   1          2       3

// nessuna colonna da visualizzare
if pmOrd = *all'0' or pmOrd = *blanks;
  return;
endif;

// scelta dell'utente per colonne da visualizzare e ordine
clear OrdCol;
OrdCol(1).Col = 'DEPTNO';
OrdCol(2).Col = 'DEPTNAME';
OrdCol(3).Col = 'MGRNO';
OrdCol(4).Col = 'ADMRDEPT';
for i = 1 to 4;
  if %subst(pmOrd:i:1) <> '0' and %subst(pmOrd:i:1) <> *blanks;
    OrdCol(i).Ord = %dec(%subst(pmOrd:i:1):1:0);
  else;
    clear OrdCol(i).Col;
    OrdCol(i).Ord = *hival;
  endif;
endfor;
sorta OrdCol(*).Ord;

// costruzione istruzione SQL
for i = 1 to %elem(OrdCol);
  if OrdCol(i).Ord <> *hival;
    if sqlstm = *blanks;
      sqlstm = 'select ' + %trim(OrdCol(i).Col);
    else;
      sqlstm = sqlstm + ', ' + %trim(OrdCol(i).Col);
    endif;
  endif;
endfor;
sqlstm = sqlstm + ' from DEPARTMENT where ADMRDEPT = ? order by DEPTNO';

// preparazione istruzione SQL
exec sql
  prepare pCurs1 from :sqlstm;

// reperisco la definizione del resultset del cursore
exec sql
  describe pCurs1 using sql descriptor local :Desc;
exec sql
  get descriptor :Desc :numCol = count;

// apertura cursore
exec sql
  open Curs1 using :pmDept;

dou sqlcode = 100;
  // leggo il cursore indirizzando l'output al descrittore
  exec sql
    fetch next from Curs1 into sql descriptor :Desc;

  if sqlcode = 100;
    leave;
  endif;
  clear csvRiga;
  for i = 1 to numCol;
    // reperimento informazioni sulla colonna (nome e tipo)
    exec sql
      get descriptor :Desc value :i :NomeCol = DB2_SYSTEM_COLUMN_NAME,
                     :TypeCol = TYPE;
    // alla lettura del primo record scrivo anche le intestazioni del file csv
    if firstRow;
      csvInt = csvInt + %trim(NomeCol) + ';';
    endif;
    // reperimento contenuto della colonna in base al tipo dati
    select;
      // numero
      when TypeCol >= 2 and TypeCol <= 8 or TypeCol = 25;
      // carattere a lunghezza fissa
      when TypeCol = 1;
        exec sql
          get descriptor :Desc value :i :wChar = DATA;
        csvRiga = csvRiga + %trim(wChar) + ';';
      // carattere a lunghezza variabile
      when TypeCol = 12;
        exec sql
          get descriptor :Desc value :i :wVarChar = DATA;
        csvRiga = csvRiga + wVarChar + ';';
    endsl;
  endfor;
  if firstRow;
    firstRow = *off;
    // scrivo su IFS csvInt
    //  (cr. Power Coffee 25/2021)
    exec sql
      call ifs_write_utf8(path_name => :FileCsv,
                     line => :csvInt,
                     overwrite => 'REPLACE',
                     end_of_line => 'CRLF');

  endif;
  // scrivo su IFS csvRiga
  exec sql
    call ifs_write_utf8(path_name => :FileCsv,
                   line => :csvRiga,
                   overwrite => 'APPEND',
                   end_of_line => 'CRLF');

enddo;

exec sql
  close Curs1;

exec sql
  deallocate descriptor local :Desc;

*inlr = *on;


**free

// ESEMPIO UTILIZZO SQL DESCRIPTOR CON SQL DINAMICO CON RESULTSET VARIABILE
// (c) MarkOne 2022, www.markonetools.it

// esempio di costruzione di una istruzione SQL dinamica in base a un
//  parametro di filtro (codice ufficio di riferimento)
//  il parametro codice ufficio di riferimento può essere blanks che equivale
//  a visualizzare tutti i record
//  l'output del resultset è variabile per numero e ordine delle colonne
// per semplificare l'esempio sono state omesse tutte le parti di controllo
//  dell'esito delle istruzioni SQL
// in questo esempio viene costruito in output un file csv

// esempio utilizzo:
//   call desc02 parm('A00' '0132')
//    parm n.1: codice ufficio di riferimento per filtrare i record della
//              tabella DEPARTMENT
//    parm n.2: ordine delle 4 colonne del resultset (DEPTNO, DEPTNAME, MGRNO, ADMRDEPT)
//              '0' per non visualizzare la colonna

// Curs1: cursore sql
// dCurs1A: descrittore sql (referenziato con variabile DescA)
//          per governare il numero di parameter marker
// dCurs1B: descrittore sql (referenziato con variabile DescB)
//          per ottenere informazioni sulla struttura del resultset
// pCurs1: istruzione preparata per il cursore Curs1 in base alla stringa sqlstm

//--- MAIN INTERFACE ---
dcl-pr DESC03 extpgm('DESC03');
 *n like(DEPTNO);
 *n char(4);
end-pr;
dcl-pi DESC03;
 pmDept like(DEPTNO);
 pmOrd char(4);
end-pi;

dcl-ds DEPARTMENT ext template end-ds;

// istruzione SQL dinamica
dcl-s sqlstm varchar(1000);
// numero massimo di elementi usati nel descrittore
dcl-s MaxEl uns(3) inz(%size(pmOrd));

// array per ordinamento colonne del resultset (parametro input)
dcl-ds OrdCol dim(%size(pmOrd)) qualified;
  Ord zoned(1);
  Col char(10);
end-ds;
// intestazioni colonne csv output
dcl-s NomeCol varchar(10);
// tipo dati della colonna di output
dcl-s TypeCol int(10);
// contenuto della colonna di output
dcl-s wVarChar varchar(100);
dcl-s wChar char(100);
// numero di colonne del resultset
dcl-s numCol uns(3);
dcl-s i uns(3);

// nome descrittore per governare il numero di parameter marker
dcl-s DescA varchar(20) inz('dCurs1A');
// nome descrittore per ottenere informazioni sulla struttura del resultset
dcl-s DescB varchar(20) inz('dCurs1B');

// boolean per primo record letto dal resultset
dcl-s firstRow ind inz(*on);

// percorso file csv di output
dcl-s FileCsv varchar(256) inz('/home/MK1sample/Department2.csv');
// riga di intestazione del csv
dcl-s csvInt varchar(500);
// altre righe del csv
dcl-s csvRiga varchar(500);

// dichiarazione cursore
exec sql
  declare Curs1 cursor for pCurs1;

//------------------------------------------------------------------------------

// allocazione descrittore SQL con massimo 1 elementi
//  ovverosia il numero massimo di parameter marker
exec sql
  allocate descriptor local :DescA with max 1;
// allocazione descrittore SQL con massimo 4 elementi
//  ovverosia il numero massimo di colonne del resultset
exec sql
  allocate descriptor local :DescB with max :MaxEl;

// -> array popolato in base a input dell'utente
// OrdCol indice 1        2          3       4
// .Col          DEPTNO _ DEPTNAME _ MGRNO _ ADMRDEPT _
// .Ord                 0          1       3          2  <-- ordinamento scelto da utente
// -> Riordino l'array in base a .Ord
// OrdCol indice 1        2          3       4
// .Col          DEPTNAME _ ADMRDEPT _ MGRNO _
// .Ord                   1          2       3

// nessuna colonna da visualizzare
if pmOrd = *all'0' or pmOrd = *blanks;
  return;
endif;

// scelta dell'utente per colonne da visualizzare e ordine
clear OrdCol;
OrdCol(1).Col = 'DEPTNO';
OrdCol(2).Col = 'DEPTNAME';
OrdCol(3).Col = 'MGRNO';
OrdCol(4).Col = 'ADMRDEPT';
for i = 1 to 4;
  if %subst(pmOrd:i:1) <> '0' and %subst(pmOrd:i:1) <> *blanks;
    OrdCol(i).Ord = %dec(%subst(pmOrd:i:1):1:0);
  else;
    clear OrdCol(i).Col;
    OrdCol(i).Ord = *hival;
  endif;
endfor;
sorta OrdCol(*).Ord;

// costruzione istruzione SQL
for i = 1 to %elem(OrdCol);
  if OrdCol(i).Ord <> *hival;
    if sqlstm = *blanks;
      sqlstm = 'select ' + %trim(OrdCol(i).Col);
    else;
      sqlstm = sqlstm + ', ' + %trim(OrdCol(i).Col);
    endif;
  endif;
endfor;
if pmDept <> *blanks;
  sqlstm = sqlstm + ' from DEPARTMENT where ADMRDEPT = ? order by DEPTNO';
  // se ho bisogno del parameter marker comunico al descrittore "come è fatto"
  //   il parametro e quale è il suo valore
  exec sql
    set descriptor :descA
      value 1 type = 1, length = 3, data = :pmDept;
  exec sql
    set descriptor :descA count = 1;
else;
  sqlstm = sqlstm + ' from DEPARTMENT order by DEPTNO';
  // se non ho bisogno del parameter marker comunico al descrittore
  //   che il numero di elementi è zero
  exec sql
    set descriptor :descA count = 0;
endif;
// preparazione istruzione SQL
exec sql
  prepare pCurs1 from :sqlstm;

// reperisco la definizione del resultset del cursore usando il descrittore dCirs1B
exec sql
  describe pCurs1 using sql descriptor local :DescB;
exec sql
  get descriptor :DescB :numCol = count;

// apertura cursore usando il descrittore dCurs1A
exec sql
  open Curs1 using SQL descriptor :descA;

dou sqlcode = 100;
  // leggo il cursore indirizzando l'output al descrittore dCurs1B
  exec sql
    fetch next from Curs1 into sql descriptor :DescB;

  if sqlcode = 100;
    leave;
  endif;
  clear csvRiga;
  for i = 1 to numCol;
    // reperimento informazioni sulla colonna (nome e tipo)
    exec sql
      get descriptor :DescB value :i :NomeCol = DB2_SYSTEM_COLUMN_NAME,
                     :TypeCol = TYPE;
    // alla lettura del primo record scrivo anche le intestazioni del file csv
    if firstRow;
      csvInt = csvInt + %trim(NomeCol) + ';';
    endif;
    // reperimento contenuto della colonna in base al tipo dati
    select;
      // numero
      when TypeCol >= 2 and TypeCol <= 8 or TypeCol = 25;
      // carattere a lunghezza fissa
      when TypeCol = 1;
        exec sql
          get descriptor :DescB value :i :wChar = DATA;
        csvRiga = csvRiga + %trim(wChar) + ';';
      // carattere a lunghezza variabile
      when TypeCol = 12;
        exec sql
          get descriptor :DescB value :i :wVarChar = DATA;
        csvRiga = csvRiga + wVarChar + ';';
    endsl;
  endfor;
  if firstRow;
    firstRow = *off;
    // scrivo su IFS csvInt
    //  (cr. Power Coffee 25/2021)
    exec sql
      call ifs_write_utf8(path_name => :FileCsv,
                     line => :csvInt,
                     overwrite => 'REPLACE',
                     end_of_line => 'CRLF');

  endif;
  // scrivo su IFS csvRiga
  exec sql
    call ifs_write_utf8(path_name => :FileCsv,
                   line => :csvRiga,
                   overwrite => 'APPEND',
                   end_of_line => 'CRLF');

enddo;

exec sql
  close Curs1;

exec sql
  deallocate descriptor local :DescA;
exec sql
  deallocate descriptor local :DescB;

*inlr = *on;

